



Techniques of Water-Resources Investigations
of the United States Geological Survey

Chapter A1

**A MODULAR THREE-DIMENSIONAL
FINITE-DIFFERENCE GROUND-WATER
FLOW MODEL**

By Michael G. McDonald and
Arlen W. Harbaugh

This chapter supersedes U.S. Geological
Survey Open-File Report 83-875

Book 6

MODELING TECHNIQUES

Chapter 6

RIVER PACKAGE

Conceptualization and Implementation

Rivers and streams contribute water to the ground-water system or drain water from it depending on the head gradient between the stream and the ground-water regime. The purpose of the River Package is to simulate the effects of flow between surface-water features and ground-water systems. To accomplish this, terms representing seepage to or from the surface features must be added to the ground-water flow equation (equation (26)) for each cell affected by the seepage.

Figure 32 shows a stream divided into reaches so that each reach is completely contained in a single cell. Stream aquifer seepage is simulated between each reach and the model cell that contains that reach.

The cross-section of figure 33-a shows a situation in which the open water of a stream is separated from the ground-water system by a layer of low permeability streambed material. Figure 33-b shows an idealization of this system in which the stream-aquifer interconnection is represented as a simple conductance through which one-dimensional flow occurs. The system of figure 33 is helpful in conceptualizing and describing the simulation of stream-aquifer interaction; however, it must be recognized that, in many instances, no discrete low-permeability streambed layer is present. The techniques of simulation developed through the conceptualization of figure 33 can still be applied to represent these situations, provided the proper interpretation is placed on the various terms and parameters that are used.

Figure 34 shows an isolated view of the idealized streambed conductance of figure 33-b, as it crosses an individual cell. The length of the




Figure 32.—Discretization of a stream into reaches. Some small reaches are ignored.






Figure 33.—(a) Cross section of an aquifer containing a stream and (b) Conceptual representation of stream-aquifer interconnection in simulation.



Streambed Conductance = KLW/M

Figure 34.—Idealization of streambed conductance in an individual cell.

conductance block is taken as the length of the stream, L, as it crosses the node; the width is taken as the stream width, W; the distance of flow is taken as the thickness, M, of the streambed layer; and the hydraulic conductivity of the streambed material is designated K. The assumption is made that measurable head losses between the stream and the aquifer are limited to those across the streambed layer itself--that is, that there is no significant head loss between the bottom of the streambed layer and the point represented by the underlying model node. It is further assumed that the underlying model cell remains fully saturated--that is, that its water level does not drop below the bottom of the streambed layer. Under these assumptions, flow between the stream and the ground-water system is given by

$$QRIV = \frac{K L W}{M} (HRIV - h_{i,j,k}) \quad (63-a)$$

or


$$QRIV = CRIV (HRIV - h_{i,j,k}) \quad (63-b)$$

where QRIV is the flow between the stream and the aquifer, taken as positive if it is directed into the aquifer; HRIV is the head in the stream; CRIV is the hydraulic conductance of the stream-aquifer interconnection (KLW/M), and $h_{i,j,k}$ is the head at the node in the cell underlying the stream reach.

If the assumption is satisfied that all significant head loss occurs across a discrete streambed layer, the application of equations (63) is straightforward. More frequently, however, equations (63) must be applied to situations in which no discrete streambed layer can be identified, or in which head losses are not restricted to those across such a layer. In these cases, the task is to formulate a single conductance term, CRIV, which can be used in (63-b) to relate flow between the stream and the depth


represented by node i,j,k to the corresponding head difference. This flow is in general a three-dimensional process, and its representation through a single conductance term can never be more than approximate. If reliable field measurements of stream seepage and associated head difference are available, they may be used to calculate an effective conductance. Otherwise, a conductance value must be chosen more or less arbitrarily and adjusted during model calibration. Certain rules can be formulated to guide the initial choice of conductance. For example, the assumed cross-sectional area of flow should normally be of the same order of magnitude as the product of channel width and stream reach length within the cell; the assumed distance of flow should not exceed the vertical interval between the streambed and node i,j,k ; and, if distinct layers can be recognized within this interval, these should normally be treated as conductances in series in formulating an equivalent conductance. In general, however, it should be recognized that formulation of a single conductance term to account for a three-dimensional flow process is inherently an empirical exercise, and that adjustment during calibration is almost always required.

Equations (63) normally provide an acceptable approximation of stream-aquifer interaction over a certain range of aquifer head values. In most cases, however, if water levels in the aquifer fall below a certain point, seepage from the stream ceases to depend on head in the aquifer. This can be visualized by returning to the concept of a discrete streambed layer. Figure 35-a shows the situation described by equations (63); water level in the aquifer is above the bottom of the streambed layer, and flow through that layer is proportional to the head difference between the stream and the aquifer. In figure 35-b, water level in the aquifer has fallen below the bottom of the streambed layer, leaving an unsaturated interval beneath that



Head at the bottom of the streambed is equal to head in the cell.

A



Head at the bottom of the streambed is equal to elevation of bottom of streambed layer.

B

Figure 35.—Cross sections showing the relation between head at the bottom of the streambed layer and head in the cell. Head in the cell is equal to the water-table elevation.

layer; if it is assumed that the streambed layer itself remains saturated, the head at its base will simply be the elevation at that point. If this elevation is designated RBOT, the flow through the streambed layer is given by

$$QRIV = CRIV (HRIV - RBOT) \quad (64)$$

where QRIV, CRIV, and HRIV are as defined for equation (63-b). Obviously, further declines in head below RBOT produce no increase in flow through the streambed layer; the flow simply retains the constant value given by equation (64), as long as head remains below RBOT. The model described herein utilizes these concepts in simulating stream-aquifer interaction--that is, flow between a stream and a node i,j,k is simulated according to the equation set

$$\begin{aligned} QRIV &= CRIV (HRIV - h_{i,j,k}), h_{i,j,k} > RBOT \\ QRIV &= CRIV (HRIV - RBOT), h_{i,j,k} \leq RBOT \end{aligned} \quad (65)$$

Figure 36 shows a graph of flow between the stream and cell i,j,k as a function of the head, $h_{i,j,k}$, as calculated using equations 65. Flow is zero when $h_{i,j,k}$ is equal to the water level in the stream, HRIV. For higher values of $h_{i,j,k}$, flow is negative, that is, into the stream; for lower values of $h_{i,j,k}$, flow is positive, that is, into the aquifer. This positive flow increases linearly as $h_{i,j,k}$ decreases, until $h_{i,j,k}$ reaches RBOT; thereafter the flow remains constant.

A relationship similar to that of equations (65) and figure (36) generally prevails in stream-aquifer interaction whether or not a discrete streambed layer is present. For example, once a break in saturation occurs between the stream and the aquifer, seepage from the stream to the aquifer must become independent of head in the aquifer. In most cases, moreover, this independence is established even before a break in saturation occurs.





Figure 36.—Plot of flow, QRIV, from a stream into a cell as a function of head, h , in the cell where RBOT is the elevation of the bottom of the streambed and HRIV is the head in the stream.

Figure 37 shows a situation in which levels in the aquifer have fallen far enough below a stream so that only a narrow saturated connection exists between the streambed and the regional water table. Examination of figure 37 will show that the head gradient in the saturated connection must be approximately unity, and that further lowering of the regional water table will not increase this gradient. Thus, once a condition similar to that in figure 37 is established, seepage from the stream is independent of further head decline in the aquifer. The situation shown in figure 37 is itself an overimplification of field conditions, which may often involve complex patterns of saturated and unsaturated material beneath the stream. In all situations, however, seepage from the stream must at some point become independent of head in the aquifer, as that head continues to decline.

If hydrologic conditions indicate that seepage from a stream will increase as the local water table elevation declines, but will reach the limiting condition illustrated in figure 37 when the water table reaches an elevation h_1 , $RBOT$ should be taken equal to h_1 . Because the vertical hydraulic gradient beneath the stream is approximately one under the conditions of figure 37, seepage from the stream into cell i,j,k is given approximately by the product KLW , where K is the vertical hydraulic conductivity of the material in the saturated column beneath the stream, and again L is the length of the stream as it crosses cell i,j,k , and W is stream width. A value of the stream-aquifer conductance term, $CRIV$, consistent with this estimated value of limiting seepage and with the selected value of $RBOT$ can be obtained by substituting KLW for $QRIV$ in equation (64) and solving for $CRIV$. This yields

$$CRIV = \frac{KLW}{HRIV - RBOT} \quad (66)$$



----- Line of
Equal Head

Figure 37.—Limiting seepage from a stream at unit hydraulic gradient

In summary, if the limiting condition of stream seepage is expected to follow the model of figure 37, RBOT should be chosen as the water table elevation at which the transition to this limiting seepage can be expected, and CRIV may be calculated from equation (66). The model simulation technique based on equations (65) should then provide a reasonable approximation to the stream-aquifer interaction.

The simplified model of stream-aquifer interaction utilized here assumes that this interaction is independent of the location of the stream reach within the cell, and that the level of water in the stream is uniform over the reach, and constant over each stress period. The latter assumption implies that conditions of flow in the stream do not vary significantly during the stress period--for example, that the stream does not go dry or overflow its banks, or that such events are of such short duration as to have no effect on stream-aquifer interaction.

Data describing each river are specified by the user for each stress period. Input consists of six entries for each river reach, specifying the layer, row, and column of the cell containing the reach, and the three parameters needed to calculate seepage--stream level or stage (HRIV), the conductance of the stream-aquifer interconnection (CRIV), and the "bottom elevation," or level at which the limiting value of stream seepage is attained (RBOT).

At the start of each iteration, terms representing river seepage are added to the flow equation for each cell containing a river reach. The choice of which river seepage equation to use, equation 63 or equation 64, is made by comparing the most recent value of head at the cell to the value

of RBOT for the reach. Since this process is done at the start of each iteration, the most current value of head (HNEW) is the value from the previous iteration. Thus, the check for which river seepage equation to use lags behind the seepage calculations by one iteration. If equation 63 is selected, the term $-CRIV$ is added to the term HCOF and the term $-CRIV*HRIV$ is added to RHS. If equation 64 is selected, the term $-CRIV (HRIV - RBOT)$ is added to the term RHS.

River Package Input

Input to the River (RIV) Package is read from the unit specified in IUNIT(4).

FOR EACH SIMULATION

RIV1AL

1. Data: MXRIVR IRIVCB
Format: I10 I10

FOR EACH STRESS PERIOD

RIV1RP

2. Data: ITMP
Format: I10

3. Data: Layer Row Column Stage Cond Rbot
Format: I10 I10 I10 F10.0 F10.0 F10.0

(Input item 3 normally consists of one record for each river reach. If ITMP is negative or zero, item 3 is not read.)

Explanation of Fields Used in Input Instructions

MXRIVR--is the maximum number of river reaches active at one time.

IRIVCB--is a flag and a unit number.

If IRIVCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL (see Output Control) is set.

If IRIVCB = 0, cell-by-cell flow terms will not be printed or recorded.

If IRIVCB < 0, river leakage for each reach will be printed whenever ICBCFL is set.

ITMP--is a flag and a counter.

If ITMP < 0, river data from the last stress period will be reused.

If ITMP \geq 0, ITMP will be the number of reaches active during the current stress period.

Layer--is the layer number of the cell containing the river reach.

Row--is the row number of the cell containing the river reach.

Column--is the column number of the cell containing the river reach.

Stage--is the head in the river.

Cond--is the riverbed hydraulic conductance.

Rbot--is the elevation of the bottom of the riverbed.

SAMPLE INPUT TO THE RIVER PACKAGE

| DATA ITEM | EXPLANATION | INPUT RECORDS | | |
|-----------|--|---------------|---|------|
| 1 | {MXRIVR, IRIVCB} | 3 | | |
| 2 | {ITMP} FOR FIRST STRESS PERIOD | 3 | | |
| 3 | {LAYER, ROW, COLUMN, STAGE, COND, RBOT} FOR FIRST REACH | 2 | 4 | 220. |
| 3 | {LAYER, ROW, COLUMN, STAGE, COND, RBOT} FOR SECOND REACH | 2 | 4 | 225. |
| 3 | {LAYER, ROW, COLUMN, STAGE, COND, RBOT} FOR THIRD REACH | 2 | 4 | 210. |
| 2 | {ITMP} FOR SECOND STRESS PERIOD | -1 | | |
| 2 | {ITMP} FOR THIRD STRESS PERIOD | -1 | | |
| 2 | {ITMP} FOR FOURTH STRESS PERIOD | 2 | | |
| 3 | {LAYER, ROW, COLUMN, STAGE, COND, RBOT} FOR FIRST REACH | 2 | 4 | 210. |
| 3 | {LAYER, ROW, COLUMN, STAGE, COND, RBOT} FOR SECOND REACH | 2 | 4 | 220. |
| 2 | {ITMP} FOR FIFTH STRESS PERIOD | 0 | | |
| 2 | {ITMP} FOR SIXTH STRESS PERIOD | -1 | | |
| | | 55 | | |
| | | 6 | 4 | 212. |
| | | 7 | 4 | 217. |
| | | 5 | 4 | 200. |
| | | | | |
| | | 5 | 4 | 200. |
| | | 6 | 4 | 212. |

Module Documentation for the River Package

The River Package (RIV1) consists of four modules, all of which are called by the MAIN program. The modules are:

- RIV1AL Allocates space for a list (RIVR) which will contain an entry for each river reach. Each entry will consist of the location of the cell containing the reach, riverhead, conductance of the riverbed, and the elevation of the bottom of the riverbed.
- RIV1RP Reads, for each river reach, the location of the cell containing the reach, riverhead, conductance of the riverbed, and elevation of the bottom of the riverbed.
- RIV1FM Adds, for each river reach, the appropriate terms to the accumulators HCOF and RHS.
- RIV1BD Calculates the rates and accumulated volume of river leakage into and out of the flow system.

Narrative for Module RIVIAL

This module allocates space in the X array to store the list of river reaches.

1. Print a message identifying the package and initialize NRIVER (number of river reaches).

2. Read and print MXRIVR (the maximum number of river reaches) and IRIVCB (the unit number for saving cell-by-cell flow terms or a flag indicating whether cell-by-cell flow terms should be printed).

3. Set LCRIVR, which will point to the first element in the river list (RIVR), equal to ISUM, which is currently pointing to the first unallocated element in the X array.

4. Calculate the amount of space needed for the river list (six values for each reach--row, column, layer, riverhead, riverbed conductance, and riverbed bottom elevation) and add it to ISUM.

5. Print the number of elements in the X array used by the River Package.

6. RETURN.

Flow Chart for Module RIV1AL

NRIVER is the number of river reaches being simulated at any given time.

MXRIVR is the maximum number of river reaches simulated.


IRIVCB is a flag and a unit number.

If IRIVCB > 0, it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL (see RIV1BD module) is set.

If IRIVCB = 0, cell-by-cell flow terms will not be printed or recorded.

If IRIVCB < 0, river leakage for each reach will be printed whenever ICBCFL is set.

LCRIVR is the location in the X array of the list of river data (RIVR).



```

SUBROUTINE RIV1AL (ISUM, LENX, LCRIVR, MXRIVR, NRIVER, IN, IOUT,
1          IRIVCB)
C
C-----VERSION 1554 12MAY1987 RIV1AL
C *****:*****
C ALLOCATE ARRAY STORAGE FOR RIVERS
C *****
C
C SPECIFICATIONS:
C -----
C -----
C
C1-----IDENTIFY PACKAGE AND INITIALIZE NRIVER.
WRITE(IOUT,1)IN
1 FORMAT(1H0,'RIV1 -- RIVER PACKAGE, VERSION 1, 9/1/87',
1' INPUT READ FROM UNIT',I3)
NRIVER=0
C
C2-----READ & PRINT MXRIVR & IRIVCB(UNIT OR FLAG FOR C-B-C FLOWS)
READ(IN,2)MXRIVR, IRIVCB
2 FORMAT(2I10)
WRITE(IOUT,3)MXRIVR
3 FORMAT(1H , 'MAXIMUM OF', I5, ' RIVER NODES')
IF(IRIVCB.GT.0) WRITE(IOUT,9) IRIVCB
9 FORMAT(1X, 'CELL-BY-CELL FLOWS WILL BE RECORDED ON UNIT', I3)
IF(IRIVCB.LT.0) WRITE(IOUT,8)
8 FORMAT(1X, 'CELL-BY-CELL FLOWS WILL BE PRINTED')
C
C3-----SET LCRIVR EQUAL TO ADDRESS OF FIRST UNUSED SPACE IN X.
LCRIVR=ISUM
C
C4-----CALCULATE AMOUNT OF SPACE USED BY RIVER LIST.
ISP=6*MXRIVR
ISUM=ISUM+ISP
C
C5-----PRINT AMOUNT OF SPACE USED BY RIVER PACKAGE.
WRITE (IOUT,4)ISP
4 FORMAT(1X,I8,' ELEMENTS IN X ARRAY ARE USED FOR RIVERS')
ISUM1=ISUM-1
WRITE(IOUT,5)ISUM1,LENX
5 FORMAT(1X,I8,' ELEMENTS OF X ARRAY USED OUT OF',I8)
IF(ISUM1.GT.LENX) WRITE(IOUT,6)
6 FORMAT(1X,' ***X ARRAY MUST BE DIMENSIONED LARGER***')
C
C7-----RETURN
RETURN
END

```

List of Variables for Module RIV1AL

| <u>Variable</u> | <u>Range</u> | <u>Definition</u> |
|-----------------|--------------|---|
| IN | Package | Primary unit number from which input for this package will be read. |
| IOUT | Global | Primary unit number for all printed output. IOUT = 6. |
| IRIVCB | Package | Flag and a unit number. > 0, unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL (see RIV1BD module) is set. = 0, cell-by-cell flow terms will not be printed or recorded. < 0, river leakage for each reach will be printed whenever ICBCFL is set. |
| ISP | Module | Number of words in the X array allocated by this module. |
| ISUM | Global | Index number of the lowest element in the X array which has not yet been allocated. When space is allocated for an array, the size of the array is added to ISUM. |
| ISUM1 | Module | ISUM-1. |
| LCRIVR | Package | Location in the X array of the first element of array RIVR. |
| LENX | Global | Length of the X array in words. This should always be equal to the dimension of X specified in the MAIN program. |
| MXRIVR | Package | Maximum number of river reaches active at any one time. |
| NRIVER | Package | Number of river reaches active during the current stress period. |

Narrative for Module RIVIRP


This module reads data to build the river list.

1. Read ITMP. ITMP is the number of river reaches or a flag indicating that river reaches specified for the previous stress period should be reused.
2. Test ITMP. If ITMP is less than zero, the river data read for the last stress period will be reused. Print a message to that effect and RETURN.
3. If ITMP is greater than or equal to zero, it is the number of reaches for this stress period. Set the number of river reaches (NRIVER) in the current stress period equal to ITMP.
4. Compare the number of river reaches (NRIVER) in the current stress period to the number specified as the maximum for the simulation (MXRIVR). If NRIVER is greater than MXRIVR, STOP.
5. Print the number of river reaches in the current stress period (NRIVER).
6. See if there are any river reaches. If there are no river reaches in the current stress period (NRIVER = 0), bypass further river processing.
7. Read and print the layer, row, column, riverhead, riverbed conductance, and the elevation of the bottom of the riverbed for each reach.
8. RETURN.

Flow Chart for Module RIV1RP

ITMP is both a flag and a counter. If it is greater than or equal to zero, it is the number of reaches to be simulated during the current stress period. If it is less than zero, it indicates that the reaches simulated in the last stress period should be simulated in the current stress period.

MXRIVR is the maximum number of reaches to be simulated.




```

SUBROUTINE RIV1RP(RIVR,NRIVER,MXRIVR,IN,IOUT)
C
C
C-----VERSION 1319 25AUG1982 RIV1RP
C *****
C READ RIVER HEAD, CONDUCTANCE AND BOTTOM ELEVATION
C *****
C
C SPECIFICATIONS:
C -----
C DIMENSION RIVR(6,MXRIVR)
C -----
C
C
C1-----READ ITMP(NUMBER OF RIVER REACHES OR FLAG TO REUSE DATA)
      READ(IN,8)ITMP
      8 FORMAT(I10)
C
C2-----TEST ITMP.
      IF(ITMP.GE.0)GO TO 50
C
C2A-----IF ITMP <0 THEN REUSE DATA FROM LAST STRESS PERIOD.
      WRITE(IOUT,7)
      7 FORMAT(1H0,'REUSING RIVER REACHES FROM LAST STRESS PERIOD')
      GO TO 260
C
C3-----IF ITMP=> ZERO THEN IT IS THE NUMBER OF RIVER REACHES
      50 NRIVER=ITMP
C
C4-----IF NRIVER>MXRIVR THEN STOP.
      IF(NRIVER.LE.MXRIVR)GO TO 100
      WRITE(IOUT,99)NRIVER,MXRIVR
      99 FORMAT(1H0,'NRIVER(',I4,') IS GREATER THAN MXRIVR(',I4,')')
C
C4A-----ABNORMAL STOP.
      STOP
C
C5-----PRINT NUMBER OF RIVER REACHES IN THIS STRESS PERIOD.
      100 WRITE(IOUT,1)NRIVER
      1 FORMAT(1H0,//1X,I5,' RIVER REACHES')
C
C6-----IF THERE ARE NO RIVER REACHES THEN RETURN.
      IF(NRIVER.EQ.0) GO TO 260
C
C7-----READ AND PRINT DATA FOR EACH RIVER REACH.
      WRITE(IOUT,3)
      3 FORMAT(1H0,15X,'LAYER',5X,'ROW',5X,'COL '
      1,' STAGE CONDUCTANCE BOTTOM ELEVATION RIVER REACH'
      2//1X,15X,80('-'))
      DO 250 II=1,NRIVER
      READ(IN,4)K,I,J,RIVR(4,II),RIVR(5,II),RIVR(6,II)
      4 FORMAT(3I10,3F10.0)
      WRITE(IOUT,5)K,I,J,RIVR(4,II),RIVR(5,II),RIVR(6,II),II
      5 FORMAT(1X,15X,I4,I9,I8,G13.4,G14.4,G19.4,I10)
      RIVR(1,II)=K
      RIVR(2,II)=I
      RIVR(3,II)=J
      250 CONTINUE
C
C8-----RETURN
      260 RETURN
      END

```

List of Variables for Module RIV1RP

| <u>Variable</u> | <u>Range</u> | <u>Definition</u> |
|-----------------|--------------|--|
| I | Module | Row number. |
| II | Module | Index for river reach. |
| IN | Package | Primary unit number from which input for this package will be read. |
| IOUT | Global | Primary unit number for all printed output. IOUT = 6. |
| ITMP | Module | Flag or number of rivers. ≥ 0 , number of rivers active during the current stress period. < 0 , same rivers active during the last stress period will be active during the current stress period. |
| J | Module | Column number. |
| K | Module | Layer number. |
| MXRIVR | Package | Maximum number of river reaches active at any one time. |
| NRIVER | Package | Number of river reaches active during the current stress period. |
| RIVR | Package | DIMENSION (6,MXRIVR), For each reach: layer, row, column, river head, riverbed conductance and elevation of bottom of riverbed. |